

Programming Languages and Compilers (CS 421)

Elsa L Gunter
2112 SC, UIUC

<http://courses.engr.illinois.edu/cs421>

Based in part on slides by Mattox Beckman, as updated by Vikram Adve and Gul Agha

9/22/15

1

Terms

- A function is in **Direct Style** when it returns its result back to the caller.
- A **Tail Call** occurs when a function returns the result of another function call without any more computations (eg tail recursion)
- A function is in **Continuation Passing Style** when it, and every function call in it, passes its result to another function.
- Instead of returning the result to the caller, we pass it forward to another function.

9/22/15

2

Terminology

- **Tail Position:** A subexpression s of expressions e , such that if evaluated, will be taken as the value of e
 - if $(x > 3)$ then $x + 2$ else $x - 4$
 - let $x = 5$ in $x + 4$
- **Tail Call:** A function call that occurs in tail position
 - if $(h\ x)$ then $f\ x$ else $(x + g\ x)$

9/22/15

3

Terminology

- **Available:** A function call that can be executed by the current expression
 - The fastest way to be unavailable is to be guarded by an abstraction (anonymous function, lambda lifted).
 - if $(h\ x)$ then $f\ x$ else $(x + g\ x)$
 - if $(h\ x)$ then $(\text{fun } x \rightarrow f\ x)$ else $(g\ (x + x))$
- ↑
Not available

9/22/15

4

CPS Transformation

- Step 1: Add continuation argument to any function definition:
 - let $f\ \text{arg} = e \Rightarrow \text{let } f\ \text{arg } k = e$
 - Idea: Every function takes an extra parameter saying where the result goes
- Step 2: A simple expression in tail position should be passed to a continuation instead of returned:
 - return $a \Rightarrow k\ a$
 - Assuming a is a constant or variable.
 - "Simple" = "No available function calls."

9/22/15

5

CPS Transformation

- Step 3: Pass the current continuation to every function call in tail position
 - return $f\ \text{arg} \Rightarrow f\ \text{arg } k$
 - The function "isn't going to return," so we need to tell it where to put the result.

9/22/15

6

CPS Transformation

- Step 4: Each function call not in tail position needs to be converted to take a new continuation (containing the old continuation as appropriate)
 - return $op\ (f\ arg) \Rightarrow f\ arg\ (fun\ r\ ->\ k(op\ r))$
 - op represents a primitive operation
- return $f(g\ arg) \Rightarrow g\ arg\ (fun\ r\ ->\ f\ r\ k)$

9/22/15

7

Example

Before:

```
let rec add_list lst =
  match lst with
  [ ] -> 0
| 0 :: xs -> add_list xs
| x :: xs -> (+) x
  (add_list xs);;
```

After:

```
let rec add_listk lst k =
  (* rule 1 *)
  match lst with
  | [ ] -> k 0 (* rule 2 *)
| 0 :: xs -> add_listk xs k
  (* rule 3 *)
| x :: xs -> add_listk xs
  (fun r -> k ((+) x r));;
(* rule 4 *)
```

9/22/15

8

Variants - Syntax (slightly simplified)


- type $name = C_1 [of\ ty_1] | \dots | C_n [of\ ty_n]$
- Introduce a type called $name$
- $(fun\ x\ ->\ C_i\ x) : ty_i -> name$
- C_i is called a **constructor**; if the optional type argument is omitted, it is called a **constant**
- Constructors are the basis of almost all pattern matching

9/22/15

9

Enumeration Types as Variants

An enumeration type is a collection of distinct values



In C and Ocaml they have an order structure; order by order of input

9/22/15

10

Enumeration Types as Variants

```
# type weekday = Monday | Tuesday | Wednesday
  | Thursday | Friday | Saturday | Sunday;;
```

```
type weekday =
  Monday
| Tuesday
| Wednesday
| Thursday
| Friday
| Saturday
| Sunday
```

9/22/15

11

Functions over Enumerations

```
# let day_after day = match day with
  Monday -> Tuesday
  | Tuesday -> Wednesday
  | Wednesday -> Thursday
  | Thursday -> Friday
  | Friday -> Saturday
  | Saturday -> Sunday
  | Sunday -> Monday;;
val day_after : weekday -> weekday = <fun>
```

9/22/15

12

Functions over Enumerations

```
# let rec days_later n day =  
  match n with 0 -> day  
  | _ -> if n > 0  
    then day_after (days_later (n - 1) day)  
    else days_later (n + 7) day;;  
val days_later : int -> weekday -> weekday  
= <fun>
```

9/22/15

13

Functions over Enumerations

```
# days_later 2 Tuesday;;  
- : weekday = Thursday  
# days_later (-1) Wednesday;;  
- : weekday = Tuesday  
# days_later (-4) Monday;;  
- : weekday = Thursday
```

9/22/15

14

Problem:

```
# type weekday = Monday | Tuesday |  
  Wednesday  
  | Thursday | Friday | Saturday | Sunday;;  
■ Write function is_weekend : weekday -> bool  
let is_weekend day =
```

9/22/15

15

Problem:

```
# type weekday = Monday | Tuesday |  
  Wednesday  
  | Thursday | Friday | Saturday | Sunday;;  
■ Write function is_weekend : weekday -> bool  
let is_weekend day =  
  match day with Saturday -> true  
  | Sunday -> true  
  | _ -> false
```

9/22/15

16

Example Enumeration Types


```
# type bin_op = IntPlusOp | IntMinusOp  
  | EqOp | CommaOp | ConsOp  
  
# type mon_op = HdOp | TlOp | FstOp  
  | SndOp
```

9/22/15

17

Disjoint Union Types

- Disjoint union of types, with some possibly occurring more than once



- We can also add in some new singleton elements

9/22/15

18

Disjoint Union Types

```
# type id = DriversLicense of int
| SocialSecurity of int | Name of string;;
type id = DriversLicense of int | SocialSecurity
of int | Name of string
# let check_id id = match id with
  DriversLicense num ->
    not (List.mem num [13570; 99999])
  | SocialSecurity num -> num < 900000000
  | Name str -> not (str = "John Doe");;
val check_id : id -> bool = <fun>
```

9/22/15

19

Problem

- Create a type to represent the currencies for US, UK, Europe and Japan

9/22/15

20

Problem

- Create a type to represent the currencies for US, UK, Europe and Japan

```
type currency =
  Dollar of int
  | Pound of int
  | Euro of int
  | Yen of int
```

9/22/15

21

Example Disjoint Union Type

```
# type const =
  BoolConst of bool
  | IntConst of int
  | FloatConst of float
  | StringConst of string
  | NilConst
  | UnitConst
```

9/22/15

22

Example Disjoint Union Type

```
# type const = BoolConst of bool
| IntConst of int | FloatConst of float
| StringConst of string | NilConst
| UnitConst
```

- How to represent 7 as a const?
- Answer: `IntConst 7`

9/22/15

23

Polymorphism in Variants

- The type `'a option` is gives us something to represent non-existence or failure

```
# type 'a option = Some of 'a | None;;
type 'a option = Some of 'a | None
```

- Used to encode partial functions
- Often can replace the raising of an exception

9/22/15

24

Functions producing option

```
# let rec first p list =
  match list with [ ] -> None
  | (x::xs) -> if p x then Some x else first p xs;;
val first : ('a -> bool) -> 'a list -> 'a option = <fun>
# first (fun x -> x > 3) [1;3;4;2;5];;
- : int option = Some 4
# first (fun x -> x > 5) [1;3;4;2;5];;
- : int option = None
```

9/22/15

25

Functions over option

```
# let result_ok r =
  match r with None -> false
  | Some _ -> true;;
val result_ok : 'a option -> bool = <fun>
# result_ok (first (fun x -> x > 3) [1;3;4;2;5]);;
- : bool = true
# result_ok (first (fun x -> x > 5) [1;3;4;2;5]);;
- : bool = false
```

9/22/15

26

Problem

- Write a hd and tl on lists that doesn't raise an exception and works at all types of lists.

9/22/15

27

Problem

- Write a hd and tl on lists that doesn't raise an exception and works at all types of lists.

```
let hd list =
  match list with [ ] -> None
  | (x::xs) -> Some x
let tl list =
  match list with [ ] -> None
  | (x::xs) -> Some xs
```

9/22/15

28

Mapping over Variants

```
# let optionMap f opt =
  match opt with None -> None
  | Some x -> Some (f x);;
val optionMap : ('a -> 'b) -> 'a option -> 'b option = <fun>
# optionMap
  (fun x -> x - 2)
  (first (fun x -> x > 3) [1;3;4;2;5]);;
- : int option = Some 2
```

9/22/15

29

Folding over Variants


```
# let optionFold someFun noneVal opt =
  match opt with None -> noneVal
  | Some x -> someFun x;;
val optionFold : ('a -> 'b) -> 'b -> 'a option -> 'b = <fun>
# let optionMap f opt =
  optionFold (fun x -> Some (f x)) None opt;;
val optionMap : ('a -> 'b) -> 'a option -> 'b option = <fun>
```

9/22/15

30

Recursive Types

- The type being defined may be a component of itself



9/22/15

31

Recursive Data Types

```
# type exp =  
  VarExp of string  
  | ConstExp of const  
  | MonOpAppExp of mon_op * exp  
  | BinOpAppExp of bin_op * exp * exp  
  | IfExp of exp * exp * exp  
  | AppExp of exp * exp  
  | FunExp of string * exp
```

9/22/15

32

Recursive Data Types

```
# type bin_op = IntPlusOp | IntMinusOp  
  | EqOp | CommaOp | ConsOp | ...  
# type const = BoolConst of bool | IntConst of int | ...  
# type exp = VarExp of string | ConstExp of const  
  | BinOpAppExp of bin_op * exp * exp | ...
```

- How to represent 6 as an exp?

9/22/15

33

Recursive Data Types

```
# type bin_op = IntPlusOp | IntMinusOp  
  | EqOp | CommaOp | ConsOp | ...  
# type const = BoolConst of bool | IntConst of int | ...  
# type exp = VarExp of string | ConstExp of const  
  | BinOpAppExp of bin_op * exp * exp | ...
```

- How to represent 6 as an exp?
- Answer: ConstExp (IntConst 6)

9/22/15

34

Recursive Data Types

```
# type bin_op = IntPlusOp | IntMinusOp  
  | EqOp | CommaOp | ConsOp | ...  
# type const = BoolConst of bool | IntConst of int | ...  
# type exp = VarExp of string | ConstExp of const  
  | BinOpAppExp of bin_op * exp * exp | ...
```

- How to represent (6, 3) as an exp?

9/22/15

35

Recursive Data Types

```
# type bin_op = IntPlusOp | IntMinusOp  
  | EqOp | CommaOp | ConsOp | ...  
# type const = BoolConst of bool | IntConst of int | ...  
# type exp = VarExp of string | ConstExp of const  
  | BinOpAppExp of bin_op * exp * exp | ...
```

- How to represent (6, 3) as an exp?
- BinOpAppExp (CommaOp, ConstExp (IntConst 6), ConstExp (IntConst 3))

9/22/15

36

Recursive Data Types

```
# type bin_op = IntPlusOp | IntMinusOp
  | EqOp | CommaOp | ConsOp | ...
# type const = BoolConst of bool | IntConst of int |
...
# type exp = VarExp of string | ConstExp of const
  | BinOpAppExp of bin_op * exp * exp | ...
```

- How to represent [(6, 3)] as an exp?
- BinOpAppExp (ConsOp, BinOpAppExp (CommaOp, ConstExp (IntConst 6), ConstExp (IntConst 3)), ConstExp NilConst));;

9/22/15

37

Your turn now

Try Problem 1 on MP3

9/22/15

38

Recursive Data Types

```
# type int_Bin_Tree =
  Leaf of int | Node of (int_Bin_Tree *
    int_Bin_Tree);;
```

type int_Bin_Tree = Leaf of int | Node of
(int_Bin_Tree * int_Bin_Tree)

9/22/15

39

Recursive Data Type Values


```
# let bin_tree =
  Node(Node(Leaf 3, Leaf 6), Leaf (-7));;
```

```
val bin_tree : int_Bin_Tree = Node (Node
  (Leaf 3, Leaf 6), Leaf (-7))
```

9/22/15

40

Recursive Data Type Values



9/22/15

41

Recursive Functions

```
# let rec first_leaf_value tree =
  match tree with (Leaf n) -> n
  | Node (left_tree, right_tree) ->
    first_leaf_value left_tree;;
```

```
val first_leaf_value : int_Bin_Tree -> int =
  <fun>
```

```
# let left = first_leaf_value bin_tree;;
val left : int = 3
```

9/22/15

42

Problem

```
type int_Bin_Tree = Leaf of int
| Node of (int_Bin_Tree * int_Bin_Tree);;
■ Write sum_tree : int_Bin_Tree -> int
■ Adds all ints in tree
let rec sum_tree t =
```

9/22/15

43

Problem

```
type int_Bin_Tree = Leaf of int
| Node of (int_Bin_Tree * int_Bin_Tree);;
■ Write sum_tree : int_Bin_Tree -> int
■ Adds all ints in tree
let rec sum_tree t =
  match t with Leaf n -> n
  | Node(t1,t2) -> sum_tree t1 + sum_tree t2
```

9/22/15

44

Mapping over Recursive Types

```
# let rec ibtreeMap f tree =
  match tree with (Leaf n) -> Leaf (f n)
  | Node (left_tree, right_tree) ->
    Node (ibtreeMap f left_tree,
          ibtreeMap f right_tree);;
val ibtreeMap : (int -> int) -> int_Bin_Tree ->
int_Bin_Tree = <fun>
```

9/22/15

45

Mapping over Recursive Types

```
# ibtreeMap ((+) 2) bin_tree;;

- : int_Bin_Tree = Node (Node (Leaf 5, Leaf
8), Leaf (-5))
```

9/22/15

46

Folding over Recursive Types

```
# let rec ibtreeFoldRight leafFun nodeFun tree =
  match tree with Leaf n -> leafFun n
  | Node (left_tree, right_tree) ->
    nodeFun
    (ibtreeFoldRight leafFun nodeFun left_tree)
    (ibtreeFoldRight leafFun nodeFun right_tree);;
val ibtreeFoldRight : (int -> 'a) -> ('a -> 'a -> 'a) ->
int_Bin_Tree -> 'a = <fun>
```

9/22/15

47

Folding over Recursive Types

```
# let tree_sum =
  ibtreeFoldRight (fun x -> x) (+);;
val tree_sum : int_Bin_Tree -> int = <fun>
# tree_sum bin_tree;;

- : int = 2
```

9/22/15

48